Agricultural Experiment Station—University of Minnesota

Miscellaneous Report 86—1969

Solar Radiation and Sunshine Duration Relationships

Donald A. Haines and Donald G. Baker

CONTENTS PAGE

Program RADCOR was written for and run on a Control Data Corporation 1604 FORTRAN computer for the purpose of correlating solar radiation data with sunshine duration records taken at the same or nearby stations. There were two objectives for developing RADCOR and undertaking this study: (1) to expand the solar radiation measuring network through the correlation of radiation data with a frequently observed climatological parameter and (2) to analyze the solar radiation records available within the north-central region and Alaska in some detail. The first objective is the subject of Solar Radiation and Sunshine Duration Relationships in the North-Central Region and Alaska: Basic Computation, University of Minnesota Agricultural Experiment Station Technical Bulletin 262 (NCR 195). The second objective will be explored in a following publication.

This bulletin was prepared to describe the computer program RADCOR. Examples of the printout tables and column heading terminology are given in the section entitled "Output Examples." An explanation of the mathematical methods used in this program and a detailed commentary on the printout data appear in the previously named bulletin.

Data Cards

The data cards were punched and verified at the National Weather Records Center, Environmental Science Services Administration, Asheville, North Carolina. Each card contained the total daily solar radiation for a parent station and sunshine and cloud data for up to six associated satellite stations. The exact card format information follows.

Columns	Information
1-5	Solar radiation station number
6-7	Year
8-9	Month
10-11	Day
12-16	Measured solar radiation value
17-20	Blank
21-24	Sunshine station one: Sunshine in minutes per day
25-27	Sunshine station one: Sunshine in percentage possible per day
28	Sunshine station one: Cloud amount
29-32	Sunshine station two: Sunshine in minutes per day
33-35	Sunshine station two: Sunshine in percentage possible per day
36	Sunshine station two: Cloud amount

Columns	Information
37-40	Sunshine station three: Sunshine in minutes per day
41-43	Sunshine station three: Sunshine in percentage possible per day
44	Sunshine station three: Cloud amount
45-48	Sunshine station four: Sunshine in minutes per day
49-51	Sunshine station four: Sunshine in percentage possible per day
52	Sunshine station four: Cloud amount
53-56	Sunshine station five: Sunshine in minutes per day
57-59	Sunshine station five: Sunshine in percentage possible per day
60	Sunshine station five: Cloud amount
61-64	Sunshine station six: Sunshine in minutes per day
65-67	Sunshine station six: Sunshine in percentage possible per day
68	Sunshine station six: Cloud amount
69-78	Blank
79-80	Climatological week number

Composition of the Program Card Deck

The program deck is prefaced by required computer installation cards that usually include an identification or project card and an execute card. These cards are followed by the RADCOR deck. The RADCOR program END card has a blank card following it. This card is required because of specific demands of the Minnesota CDC 1604 compiler. The program also requires three cards containing the total extraterrestrial radiation for the 52 (plus partial 53) climatological weeks of the year. The format calls for four card columns to a week, 20 weeks to a card. Since these are climatological weeks, the year begins with March 1-7 as week 1. Next in the deck makeup is a single card containing NNN in columns 1 and 2; i.e., the number of sunshine stations to be used in this run. NNN may vary from one to six. A card with radiation station documentation information is next. It is followed by from one to six cards (NNN) containing individual sunshine station documentation. Any number of data cards follows. Each card contains the information described in the section entitled "Data Cards." Either a blank card or a card containing zero punches in columns 79 and 80 follows. When this card has been read in, all data cards have been processed.

The Program

So far as possible, program RADCOR was designed to be run on any large computer that has a FORTRAN compiler. Two sections probably would have to be changed if the run is not done specifically on a CDC 1604. Two tests on the card data are made quite early in the program. One test examines overpunches (representing 10/10 cloudiness) or blanks in the cloud column. The other test is for blanks in the sunshine-minutes columns. Changes in these sections will depend upon the number of bits that make up a fixed word in a given machine.

The nomenclature used in the program, the program itself, and the flow diagram follow.

Nomenclature

A	The regression term a
AA Terms	All except AASQ are used to get averages for the year; i.e., the annual figures.
AAA	The 52 week average of A
AAB	The 52 week average of B
AACLD	The 52 week average of CLOUD
AACOR	The 52 week average of CORR
AACSQ	The 52 week average of CORRSQ
AAEXT	The 52 week average of TOEXT
AAPSN	The 52 week average of PCSUN
AARTN	The 52 week average of RADTN
AASQ	Along with PPSUN and RPDTN, it is a factor in the computation of the sums of squares.
AATR	The 52 week average of TORAD
AN	The number of cases; the count is in floating point.
B	The regression term b
BY	From the regression computation; used for the regression of y on x.
C	The cloud cover in floating point
CARDSQ	The corrected sums of squares for the ratio of measured radiation to extraterrestrial radiation
CC	Ratio of measured radiation to exterrestrial radiation

Nomenclature

CCNF	An intermediate term used in computing confidence limits
CHECK	A variable used to increment values
CLOUD	The sum and average of cloud cover (C)
COMPR	The predicted mean radiation for each 10 percent increment of measured sunshine
CONF	Ninety-five percent confidence limits for predicted mean radiation
CORR	The correlation coefficient
CORRSQ	The correlation coefficient squared
CRADSQ	A value obtained in the computation of RADSQ
D	The cloud cover; up to six stations per card read in under an R format.
FREQS	Classification to compute chance of having at least a given percentage of possible sunshine
FREQR	Classification to compute chance of having at least a given total of langleys
MWEEK	The climatological week number; when MWEEK is blank, the data cards have all been read.
NNN	The number of sunshine stations, from one to six
PCSUN	Recorded percentage of possible sunshine
PER	A variable used to increment values
PPSUN	Along with AASQ and RPDTN, it is a factor in the computation of squares.
RAD	The measured radiation value (R)
RADSQ	The sum of radiation squares CC*CC
RADTN	The sum of CC
RPDTN	A term used to compute RADSQ. Along with PPSUN and AASQ, it is a factor in the computation of sums of squares.
RSTAT	The documentation information (for the radiation station) obtained from a special data card
S	Sunshine in minutes; up to six stations per card read in under an R format.
SM	Sunshine in minutes, floating point

Nomenclature

SNRDSQ	An intermediate term used in the computation of confidence limits
SP	Sunshine in percentage, up to six stations per card
SSTAT	The documentation information (for sunshine stations) obtained from special data cards; there can be up to six cards per run.
SUNRAD	The sums of cross products SP*CC
SUNSQ	The sum of squares of SP*SP
SZ	An intermediate term used in computing FREQS
TOEXT	Total extraterrestrial radiation (R*); it is read into the computer on three data cards.
TORAD	Cumulative daily measured solar radiation totaled during a given climatological week
TOSUN	Cumulative daily sunshine in minutes totaled during a given climatological week

```
      PROGRAM RADCOR
C PROGRAM WAS WRITTEN FOR THE CDC 1604 COMPUTER.
C HOWEVER, IT IS DESIGNED AS A UNIVERSAL PROGRAM TO BE RUN ON ANY
C FORTRAN COMPILER WITH A FEW MODIFICATIONS.
C
      DIMENSION TOEXT(53),TORAD(53),TOSUN(6,53),PCSUN(6,53),CLOUD(6,53),
     1RSTAT(10),          RADTN(53),RADSQ(53),AN(53) ,SM(6),SP(6),C(6),
     2SUNRAD(6,53),SUNSQ(6,53),AATOS(6),AAPSN(6),AACLD(6),AACOR(6),
     3AACSQ(6),AAA(6),AAB(6),AASQ(6,53),RPDTN(53),PPSUN(6,53),CORR(6,53)
     4,CORRSQ(6,53),A(6,53),B(6,53),BY(6,53),COMPR(11),CONF(11),SNRDSQ
     5(6,53),SSTAT(6,10),D(6),FREQS(6,53,9),FREQR(53,8),S(6)
      TYPE INTEGER D
      TYPE INTEGER S
C
C THE FOLLOWING SECTIONS INITIALIZE THE PROGRAM.
C
      DO 1034 IA= 1,2862
 1034 FREQS (IA)=0.0
      DO 1035 IA=1,424
 1035 FREQR (IA)=0.0
      DO 1036 IA=1,48
 1036 AATOS (IA)=0.0
      DO 1032 IA=1,318
      TOSUN(IA)=0.0
      PCSUN(IA)=0.0
      CLOUD(IA)=0.0
      SUNRAD(IA)=0.0
 1032 SUNSQ(IA)=0.0
      DO 1033 IA=1,53
      AN(IA)=0.0
      RADTN(IA) =0.0
      RADSQ(IA)=0.0
 1033 TORAD(IA)=0.0
C
C INITIALIZING IS COMPLETE.
C
C TOEXT COMES IN BY USAGE OF 3 CARDS AT THE BEGINNING OF THE DATA DECK.
C IT IS THE EXTERRESTRIAL RADIATION.
C
      READ 10,(TOEXT(J),J=1,53)
   10 FORMAT(20F4.0)
C
C THIS SYMBOL, NNN, SPECIFIES THE NUMBER OF SUNSHINE STATIONS INVOLVED
C IN THIS RUN.
C
      READ 11,NNN
 11   FORMAT (I2)
C
C RSTAT AND SSTAT CARDS SPECIFY DOCUMENTATION INFORMATION.
C
      READ 50,(RSTAT(I),I=1,10)
      DO 52 I=1,NNN
   50 FORMAT(10A8)
   52 READ 50,(SSTAT(I,K),K=1,10)
C
C AT THIS POINT THE COMPUTER BEGINS TO READ REGULAR DATA CARDS
C GIVING RADIATION, SUNSHINE, AND CLOUD DATA.
C
    2 READ 3,   RAD,S (1),SP(1),D(1),S (2),SP(2),D(2),S (3),SP(3),D(3),
     1S (4),SP(4),D(4),S (5),SP(5),D(5),S (6),SP(6),D(6),MWEEK
    3 FORMAT(12X,  F4.1,4X,R4  ,F3.2,R1  ,5(R4  ,F3.2,R1  ),10X,I2)
```

```
C
C WHEN THE COMPUTER FINDS BLANKS IN THE MWEEK COLUMNS, THE
C DATA HAVE ALL BEEN READ IN.
C IF THE RADIATION COLUMNS ARE BLANK, THE COMPUTER REJECTS THE CARD.
C
      IF (MWEEK)6,6,5
    5 IF (RAD)1,2,1
    1 J=MWEEK
C
C NEXT TEST FOR OVERPUNCH IN CLOUD COLUMNS.
C IF AN OVERPUNCH IS PRESENT, THE COMPUTER ASSIGNS A 10/10 VALUE
C TO CLOUD COVER.
C THE METHODS EMPLOYED IN THIS DO LOOP WILL NOT BE VALID ON ALL
C COMPILERS.
C
      DO 831 I=1,NNN
      IF(D(I).EQ.12B) 810,1020
 1020 IF (D(I).EQ. 40B)  820,1022
 1022 IF(D(I).EQ.20B) 810,830
 810  C(I)=0.0
      GO TO 831
 820  C(I)=10.0
      GO TO 831
 830  C(I)=D(I)
      GO TO 831
  831 CONTINUE
C
C TEST COMPLETED FOR OVERPUNCHING IN CLOUD COLUMNS.
C
C THE FOLLOWING DO LOOP CHECKS THE SUNSHINE COLUMNS.  IF THEY
C  ARE BLANKS THE CARD IS REJECTED  THE METHOD EMPLOYED IN THIS
C LOOP WILL NOT BE VALID ON  ALL COMPILERS.
C
 306  DO 305 KI=1,NNN
      IF (S(KI). EQ. 20202020B) 2,839
 839   DECODE (1,6667,S(KI)) SM(KI)
 6667 FORMAT(4X,F4.1)
 305  CONTINUE
 304  TORAD (J)=RAD + TORAD (J)
      CC = RAD/TOEXT(J)
      RADTN(J) = RADTN(J) + CC
      RADSQ(J) = CC*CC + RADSQ(J)
      AN(J)=AN(J) + 1.
      DO 7 I=1,NNN
      TOSUN(I,J)=SM(I) + TOSUN(I,J)
      PCSUN(I,J)=SP(I) + PCSUN(I,J)
 16    CLOUD(I,J) = C(I)+CLOUD (I,J)
      SUNRAD(I,J) =SP(I)*CC + SUNRAD(I,J)
    7 SUNSQ(I,J) = SP(I)*SP(I) + SUNSQ(I,J)
C
C SUNSHINE AND RADIATION FREQUENCY SUMMATION FOLLOWS.
C
      DO 259 I=1,NNN
      CHECK =10.
      DO 261 K=1,9
      SZ=SP(I)*100.
      IF(SZ-CHECK) 259,250,250
  250 FREQS(I,J,K)=FREQS(I,J,K)+ 1.0
 255  CHECK=CHECK +10.
 261  CONTINUE
 259  CONTINUE
```

```
  260 CHECK=100.0
      DO 258  K=1,8
      IF(RAD-CHECK) 258,257,257
  257 FREQR(J,K)=FREQR(J,K)+1.0
  258 CHECK =CHECK +100.0
C****** SUNSHINE AND RADIATION FREQUENCIES ARE SUMMED.*****
      GO TO 2
C
C COMPUTATION OF THE REGRESSION TERMS FOLLOW.
C
    6 DO 19 I=1,NNN
      DO 19 J=1,53
      B(I,J)=(AN(J)*SUNRAD(I,J)-RADTN(J)*PCSUN(I,J))/(AN(J)*SUNSQ(I,J)
     1-(PCSUN(I,J)*PCSUN(I,J)))
   19 A(I,J)=(RADTN(J)-B(I,J)*PCSUN(I,J))/AN(J)
C
C THIS DO LOOP COMPUTES THE CORRELATION COEFFICIENTS.
C
      DO 21 I=1,NNN
      DO 21 J=1,53
      BY(I,J)=(AN(J)*SUNRAD(I,J)-RADTN(J)*PCSUN(I,J))/ (AN(J)*RADSQ(J)
     1-RADTN(J)*RADTN(J))
      CORRSQ(I,J)=BY(I,J)*B(I,J)
   21 CORR(I,J)= SQRTF(CORRSQ(I,J))
C
C CORRELATION COMPUTATIONS ARE DONE.
C THE FOLLOWING DO LOOPS AVERAGE THE TERMS THAT ARE OUTPUTTED IN TABLE
C ONE.
C
      DO 22 I=1,NNN
      DO 22 J=1,53
      AASQ(I,J)= RADTN(J)*PCSUN(I,J)/AN(J)
      PPSUN(I,J)=PCSUN(I,J)**2/AN(J)
      PCSUN(I,J)=(PCSUN(I,J)/AN(J))*100.
      TOSUN(I,J) = TOSUN(I,J)/AN(J)
        TOSUN(I,J) = TOSUN(I,J)*10.0
   22 CLOUD(I,J) =(CLOUD(I,J)/AN(J))*10.
      DO 1111 J=1,53
      TORAD(J)=TORAD(J)/AN(J)
      RPDTN(J)=RADTN(J)**2/ AN(J)
      RADTN(J)=RADTN(J)/AN(J)
 1111 CONTINUE
      DO 25 I=1,NNN
      DO 25 J=1,52
      AATOS(I)=TOSUN(I,J) + AATOS(I)
      AAPSN(I)=PCSUN(I,J) + AAPSN(I)
      AACLD(I)=CLOUD(I,J) + AACLD(I)
      AACOR(I)=CORR(I,J) + AACOR(I)
      AACSQ(I)=CORRSQ(I,J) + AACSQ(I)
      AAA(I)=A(I,J) + AAA(I)
   25 AAB(I)=B(I,J) + AAB(I)
      DO 31 J=1,52
      AAEXT = TOEXT(J) + AAEXT
      AATR=TORAD(J) + AATR
   31 AARTN=RADTN(J) + AARTN
      AATR =AATR/52.
      AARTN=AARTN/52.
      AAEXT = AAEXT/52.
      DO 26 I=1,NNN
      AATOS(I)=AATOS(I)/52.
      AAPSN(I)=AAPSN(I)/52.
```

```
      AACLD(I)=AACLD(I)/52.
      AACOR(I)=AACOR(I)/52.
      AACSQ(I)=AACSQ(I)/52.
      AAA(I)=AAA(I)/52.
   26 AAB(I)=AAB(I)/52.
   71 DO 23 I=1,NNN
      PRINT 130
  130 FORMAT(1H1,3X, 98HWEEK  EXTRA.      MEASD.   R/R*      SUNSHINE
     1 CLOUD        REGRESSION        CORREL.      CORREL.,/,4X,102HNO.
     2RAD.(R*)    RAD(R)           MIN.   PRCNT.   PRCNT.         EQUATION
     3        COEFF.    COEFF. SQRD.,/,3X,102H----   --------    ------
     4  ----   ----   ------   ------    ----------------    -------   -
     5----------)
      DO 96 J=1,53
   96 PRINT 131,J,TOEXT(J),TORAD(J),RADTN(J),TOSUN(I,J),PCSUN(I,J),CLOUD
     1(I,J),A(I,J),B(I,J),CORR(I,J),CORRSQ(I,J)
  131 FORMAT(4X,I2,5X,F5.0,6X,F5.0,2X,F5.2,4X,F4.0,4X,F4.0,4X,F5.0,5X,5H
     1R=R*(,F4.3,1H+,F4.3,2HS),4X,F4.2,8X,F4.2)
      PRINT 170,AAEXT,AATR, AARTN,AATOS(I),AAPSN(I),AACLD(I),AAA(I),
     1AAB(I),AACOR(I),AACSQ(I)
  170 FORMAT(1HO,2X, 6HANNUAL,F7.0,6X,F5.0,2X,F5.2,4X,F4.0,4X,F4.0,4X,
     1F4.0,6X,5HR=R*(,F4.3,1H+,F4.3,2HS),4X,F4.2,8X,F4.2,//)
      PRINT 171,(RSTAT(K),K=1,10)
   23 PRINT 171,(SSTAT(I,K),K=1,10)
  171 FORMAT(3X,10A8)
C
C TABLE ONE IS NOW COMPLETE FOR ALL STATIONS CONSIDERED IN THIS RUN.
C
   72 DO 27 L=1,NNN
      PRINT 135
  135 FORMAT(1H1,3X,112HWEEK       NO.      MEAN      SUN.      PREDICTED ME
     1AN RADIATION FOR EACH TEN PERCENT INCREMENT OF MEASURED SUNSHINE,/
     2,3X, 111H NO.      OBS.      RAD.     PRCNT.       0       10      20
     3  30      40      50      60      70      80      90      100,/,3X,111H--
     4--      ----      ----     ------    ----------------------------------
     5-----------------------------------------)
   73 DO 140 J=1,53
      PER=0.
   74 DO 141 I=1,11
      COMPR(I) =TOEXT(J)*(A(L,J)+B(L,J)*PER)
  141 PER=PER+ 0.1
  140 PRINT 144,J,AN(J),TORAD(J),PCSUN(L,J),(COMPR(I),I=1,11)
  144 FORMAT(3X,I2,6X,F4.0,5X,F4.0,4X,F4.0,6X,F4.0,1X, 9F7.0,F7.0)
      PRINT 145,(RSTAT(K),K=1,10)
  145 FORMAT(1HO,3X,10A8)
   27 PRINT 146,(SSTAT(L,K),K=1,10)
  146 FORMAT(4X,10A8)
C
C TABLE TWO IS NOW COMPLETE FOR ALL STATIONS CONSIDERED IN THIS RUN.
C NOW FIND THE CONFIDENCE LIMITS FOR LINEAR PREDICTED RADIATION,
C AND OUTPUT THE INFORMATION THROUGH TABLE THREE.
C
   75 DO 213 L=1,NNN
   76 DO 213 J=1,53
  213 PCSUN(L,J)=PCSUN(L,J)/100.
   86 DO 28 L=1,NNN
      PRINT 150
  150 FORMAT(1H1,3X,115HWEEK   95 PERCENT CONFIDENCE LIMITS FOR PREDICTE
     1D MEAN RADIATION FOR EACH 10 PERCENT INCREMENT OF MEASURED SUNSHIN
     2E ,/,3X,112H NO.        0        10        20        30        40
     3     50        60        70        80        90        100,/,3X,11
```

```
     45H----   ------------------------------------------------------------
     5-------------------------------------------------)
  87  DO 160 J=1,53
      PER=0.
      SUNSQ(L,J)=SUNSQ(L,J)-PPSUN(L,J)
      CRADSQ=RADSQ(J)-RPDTN(J)
      SNRDSQ(L,J)=SUNRAD(L,J)-AASQ(L,J)
      CCNF=(CRADSQ-(SNRDSQ(L,J)**2)/SUNSQ(L,J))/ (AN(J)-2.)
  77  DO 165  I=1,11
      CONF(I)=2.*TOEXT(J)*(SQRTF(CCNF*(1.+1./ AN(J)+(PER-PCSUN(L,J))
     1**2 / SUNSQ(L,J))))
 165  PER=PER+ .1
 160  PRINT 172,J,(CONF(I),I=1,11)
 172  FORMAT(4X,I2,7X,F4.0,4X,F4.0,8F10.0,7X,F4.0)
      PRINT 145,(RSTAT(K),K=1,10)
  28  PRINT 146,(SSTAT(L,K),K=1,10)
  88  DO 214 L=1,NNN
      DO 214 J=1,53
 214  PCSUN(L,J)=PCSUN(L,J)*100.
C
C THE CONFIDENCE LIMITS ARE DONE.
C TABLE THREE IS NOW COMPLETE FOR ALL STATIONS CONSIDERED IN THIS RUN.
C FREQUENCY COMPUTATIONS FOLLOW FOR SUNSHINE AND RADIATION VALUES.
C
C
C
  78  DO 181 I= 1,NNN
      DO 181 J= 1,53
      DO 181 K= 1,9
 181  FREQS(I,J,K)=(FREQS(I,J,K)/AN(J))*100.
      DO 182 J=1,53
      DO 182 K=1,8
 182  FREQR(J,K)=(FREQR(J,K)/AN(J))*100.
  79  DO 187 L=1,NNN
      PRINT 188
 188  FORMAT(1H1,2X,110HCHANCE OF HAVING AT LEAST THE INDICATED PERCENT
     1POSSIBLE SUNSHINE****AND AT LEAST INDICATED TOTALS OF LANGLEYS)
      PRINT 189
 189  FORMAT(1H ,1X)
      PRINT 190
 190  FORMAT(1H ,1X,5HWEEK ,109H10  20  30  40  50  60  70  80  90 (PERC
     1ENT ACCUMULATED)XXXX 100  200  300  400  500  600  700  800(LANGLE
     2YS))
      PRINT 191
 191  FORMAT(1H ,1X,113H-----------------------------------------------
     1-------------------------------------------------------------)
      DO 192 J=1,53
 192  PRINT 193,J,(FREQS(L,J,K),K=1,9),(FREQR(J,K),K=1,8)
 193  FORMAT(1H ,2X,I2,   9F4.0,25X,8F5.0)
      PRINT 145,(RSTAT(K),K=1,10)
 187  PRINT 146,(SSTAT(L,K),K=1,10)
C
C TABLE FOUR IS NOW COMPLETE FOR ALL STATIONS CONSIDERED IN THIS RUN.
C
C THE COMPUTER RUN IS NOW COMPLETE.
      END
      END
```

Flow chart for program RADCOR

Numbers in circles refer to statement numbers in the program.



Section ii


1
From section i
Designate week number = J.
2
To section i
Cloud columns:
Overpunch?
Blank?
Numeric?
Overpunch
Set cloud cover to 10.
Numeric
No change.
Blank
Set to zero.
306
Are any sunshine (total minutes) columns blank (one to NNN)?
Yes
No
Add values from each data card to the accumulating sums for the given climatological week. These values will be used to obtain the totals for necessary computational terms.
Determine the proper numerical category for each sunshine and radiation value. This information will be used later, after averaging, in the fourth output table.


6
From section i
Compute regression terms, a and b, for all weeks, all stations (one to NNN).
Switch regression from X on Y to Y on X. Compute for all weeks, all stations (one to NNN).
Compute correlation coefficients for all weeks, all stations (one to NNN).
Compute averages for all weeks, all stations (one to NNN): sunshine percentage and its square, sums of squared terms, sunshine terms, clouds, and measured radiation and its square.
Compute yearly sums and averages (annual average) for all terms in the above box plus correlation and its square, a and b regression terms.
71
Print out weekly and yearly information for the first table.
Print out radiation and sunshine station documentation.
Is there a printout for one to NNN sunshine stations?
No
Yes
72

Section iv


72
From section iii
73
Compute COMPR, the predicted mean radiation for each 10 percent increment of measured sunshine.
Print out weekly information for the second table.
No
Are all weeks printed out?
Yes
72
Print out radiation and sunshine station documentation.
No
Is there a printout for one to NNN sunshine stations?
Yes
75


75
From section iv
Divide all sunshine percentages by 100.0.
86
Print out format (third table) for one sunshine station.
87
Compute weighted values of sunshine and radiation squared terms and their crossproducts.
77
Compute CONF, the confidence limit for predicted radiation expressed at the 95 percent level.
Are the 11 increments of the confidence limits of predicted radiation complete?
No
77
Yes
Print out a week's information for the third table.
87
Are all weeks printed?
No
Yes
Print out radiation and sunshine station documentation.
86
Is there a printout for one to NNN sunshine stations?
No
Yes
88

Section vi


88
From section v
Multiply all sunshine percentages by 100.0.
78
Obtain the averages (dependent upon the number of cases) for tables that give the chance of having at least the indicated percentage of possible sunshine.
Is the above table computed for one to NNN sunshine stations?
No
Yes
Obtain the averages (dependent upon the number of cases) for a table that gives the chance of having at least the indicated totals of langleys.
79
Print out weekly information for the fourth table.
Print out radiation and sunshine station documentation.
Is there a printout for one to NNN sunshine stations?
No
Yes
END
RADCOR

Output Examples

Four pages were printed through computer output for each solar radiation station and a given satellite sunshine station. These four pages contain the following information for each climatological week.

Page a

Mean daily extraterrestrial radiation (R*)

Mean daily measured solar radiation (R)

Ratio of mean daily measured radiation to the mean daily extraterrestrial radiation (R/R*)

Mean daily minutes of measured sunshine (MIN.)

Mean daily percentage of possible sunshine (PRCNT.)

Mean daily cloud cover in percentage from sunrise to sunset (CLOUD PRCNT.)

Regression equation of measured radiation (R) versus percentage of possible sunshine (S) (REGRESSION EQUATION)

Correlation coefficient (CORREL. COEFF.)

Correlation coefficient squared (CORREL. COEFF. SQRD.)

Page b

Total number of daily observations (NO. OBS.)

Mean daily measured solar radiation (MEAN RAD.)

Mean daily percentage of possible sunshine (SUN. PRCNT.)

Predicted mean solar radiation for each 10 percent increment of measured sunshine

Page c

The 95 percent confidence limits for the predicted mean solar radiation for each 10 percent increment of measured sunshine

Page d

Chance of having at least the indicated percentage of possible sunshine (in 10 percent increments)

Chance of having at least the indicated totals of measured solar radiation (in 100 langley increments)

Production running time on the 1604 computer averaged about 0.08 hour for a given radiation station with all accompanying sunshine stations. An example of a typical 4-page output follows. The data include the Lemont, Illinois, solar radiation record (station X5023) and the Chicago, Illinois, sunshine record (station 14819). The output covers the period from January 1957 through June 1966.

WEEK NO.	EXTRA. RAD.(R*)	MEASD. RAD(R)	R/R*	SUNSHINE MIN.	SUNSHINE PRCNT.	CLOUD PRCNT.	REGRESSION EQUATION	CORREL. COEFF.	CORREL. COEFF. SQRD.
1	589.	267.	.45	300.	44.	76.	R=R*(.201+.573S)	.89	.80
2	634.	273.	.43	281.	40.	79.	R=R*(.225+.512S)	.82	.68
3	678.	334.	.49	389.	54.	68.	R=R*(.190+.561S)	.85	.71
4	721.	338.	.47	401.	54.	72.	R=R*(.192+.512S)	.84	.70
5	763.	352.	.46	406.	53.	72.	R=R*(.160+.566S)	.90	.81
6	804.	354.	.44	378.	48.	71.	R=R*(.134+.633S)	.94	.89
7	841.	418.	.50	480.	60.	63.	R=R*(.180+.528S)	.79	.63
8	876.	416.	.47	449.	55.	68.	R=R*(.168+.560S)	.93	.86
9	908.	441.	.49	502.	60.	63.	R=R*(.120+.610S)	.93	.86
10	936.	478.	.51	571.	67.	63.	R=R*(.118+.587S)	.91	.82
11	961.	475.	.49	566.	65.	60.	R=R*(.093+.615S)	.95	.90
12	982.	490.	.50	565.	64.	60.	R=R*(.166+.521S)	.92	.84
13	998.	538.	.54	625.	70.	61.	R=R*(.179+.515S)	.83	.69
14	1011.	551.	.55	605.	67.	64.	R=R*(.251+.440S)	.82	.67
15	1019.	518.	.51	559.	62.	65.	R=R*(.186+.523S)	.92	.86
16	1023.	570.	.56	646.	71.	52.	R=R*(.176+.537S)	.90	.81
17	1022.	596.	.58	659.	72.	51.	R=R*(.281+.418S)	.82	.68
18	1017.	554.	.54	622.	67.	53.	R=R*(.281+.394S)	.73	.54
19	1008.	553.	.55	630.	70.	53.	R=R*(.190+.514S)	.87	.76
20	995.	510.	.51	568.	63.	60.	R=R*(.194+.503S)	.80	.64
21	978.	554.	.57	617.	70.	49.	R=R*(.260+.438S)	.75	.57
22	957.	488.	.51	546.	63.	60.	R=R*(.187+.514S)	.89	.80
23	932.	496.	.53	532.	62.	59.	R=R*(.271+.418S)	.86	.75
24	904.	490.	.54	531.	63.	57.	R=R*(.309+.368S)	.78	.61
25	872.	469.	.54	547.	67.	52.	R=R*(.226+.469S)	.89	.80
26	838.	438.	.52	508.	63.	57.	R=R*(.242+.445S)	.84	.70
27	802.	426.	.53	508.	65.	55.	R=R*(.201+.511S)	.91	.83
28	763.	416.	.54	498.	65.	51.	R=R*(.216+.507S)	.91	.82
29	722.	368.	.51	426.	57.	58.	R=R*(.231+.489S)	.91	.82
30	680.	302.	.44	308.	42.	70.	R=R*(.235+.496S)	.86	.75
31	637.	369.	.58	523.	74.	42.	R=R*(.219+.487S)	.86	.74
32	594.	307.	.52	403.	59.	53.	R=R*(.249+.460S)	.79	.63
33	552.	313.	.57	465.	69.	40.	R=R*(.277+.419S)	.89	.79
34	511.	231.	.45	313.	48.	64.	R=R*(.190+.548S)	.88	.77
35	472.	212.	.45	303.	48.	63.	R=R*(.205+.510S)	.93	.86
36	435.	178.	.41	263.	43.	66.	R=R*(.187+.521S)	.92	.85
37	402.	178.	.44	293.	49.	59.	R=R*(.192+.517S)	.91	.83
38	373.	125.	.34	162.	28.	80.	R=R*(.168+.605S)	.91	.83
39	349.	150.	.43	262.	46.	65.	R=R*(.172+.561S)	.92	.84
40	330.	137.	.42	252.	45.	69.	R=R*(.202+.477S)	.89	.78
41	316.	129.	.41	201.	36.	70.	R=R*(.205+.559S)	.88	.77
42	309.	147.	.47	274.	50.	61.	R=R*(.214+.522S)	.87	.76
43	307.	122.	.40	189.	34.	75.	R=R*(.218+.530S)	.87	.76
44	312.	135.	.43	224.	40.	68.	R=R*(.209+.555S)	.90	.80
45	323.	149.	.46	252.	45.	65.	R=R*(.215+.547S)	.90	.81
46	340.	166.	.49	280.	49.	66.	R=R*(.197+.587S)	.91	.82
47	362.	180.	.50	254.	44.	67.	R=R*(.258+.547S)	.85	.73
48	390.	194.	.50	262.	44.	71.	R=R*(.271+.510S)	.88	.77
49	422.	206.	.49	262.	43.	72.	R=R*(.261+.526S)	.88	.77
50	458.	200.	.44	239.	38.	70.	R=R*(.219+.567S)	.91	.82
51	498.	273.	.55	323.	50.	61.	R=R*(.266+.563S)	.92	.86
52	540.	231.	.43	220.	33.	74.	R=R*(.264+.491S)	.84	.71
53	566.	306.	.54	366.	54.	53.	R=R*(.191+.642S)	.92	.85
ANNUAL	682.	342.	.49	412.	55.	63.	R=R*(.210+.517S)	.87	.77

RADIATION STATION LEMONT, ILL. (NO. X5023) JANUARY 1957-JUNE 1966
SUNSHINE STATION CHICAGO, ILL. (NO. 14819)

WEEK NO.	NO. OBS.	MEAN RAD.	SUN, PRCNT.	PREDICTED MEAN RADIATION FOR EACH TEN PERCENT INCREMENT OF MEASURED SUNSHINE										
				0	10	20	30	40	50	60	70	80	90	100
1	59.	267.	44.	118.	152.	186.	220.	253.	287.	321.	355.	389.	422.	456.
2	63.	273.	40.	143.	175.	208.	240.	273.	305.	338.	370.	403.	435.	468.
3	63.	334.	54.	129.	167.	205.	243.	281.	319.	357.	395.	433.	471.	509.
4	63.	338.	54.	138.	175.	212.	249.	286.	323.	360.	397.	434.	471.	508.
5	67.	352.	53.	122.	165.	208.	251.	294.	338.	381.	424.	467.	510.	553.
6	70.	354.	48.	108.	159.	210.	261.	311.	362.	413.	464.	515.	566.	616.
7	70.	418.	60.	151.	196.	240.	284.	329.	373.	418.	462.	507.	551.	595.
8	70.	416.	55.	147.	196.	245.	294.	343.	393.	442.	491.	540.	589.	638.
9	67.	441.	60.	109.	165.	220.	275.	331.	386.	441.	497.	552.	608.	663.
10	63.	478.	67.	110.	165.	220.	275.	330.	385.	440.	495.	550.	605.	660.
11	62.	475.	65.	89.	149.	208.	267.	326.	385.	444.	503.	563.	622.	681.
12	63.	490.	64.	163.	214.	265.	316.	367.	419.	470.	521.	572.	623.	674.
13	63.	538.	70.	179.	230.	282.	333.	384.	436.	487.	539.	590.	641.	693.
14	68.	551.	67.	253.	298.	342.	387.	431.	476.	520.	565.	609.	654.	698.
15	70.	518.	62.	190.	243.	296.	350.	403.	456.	510.	563.	616.	670.	723.
16	70.	570.	71.	180.	235.	290.	345.	400.	455.	510.	565.	620.	675.	730.
17	70.	596.	72.	287.	330.	373.	415.	458.	501.	543.	586.	629.	672.	714.
18	66.	554.	67.	286.	326.	366.	406.	446.	486.	527.	567.	607.	647.	687.
19	63.	553.	70.	192.	243.	295.	347.	399.	451.	503.	555.	606.	658.	710.
20	63.	510.	63.	193.	243.	293.	343.	393.	443.	493.	543.	593.	643.	693.
21	63.	554.	70.	254.	297.	340.	383.	426.	469.	511.	554.	597.	640.	683.
22	63.	488.	63.	179.	228.	278.	327.	376.	425.	475.	524.	573.	622.	672.
23	63.	496.	62.	253.	292.	331.	370.	409.	448.	487.	526.	564.	603.	642.
24	62.	490.	63.	280.	313.	346.	379.	413.	446.	479.	512.	546.	579.	612.
25	63.	469.	67.	197.	238.	279.	320.	361.	402.	443.	483.	524.	565.	606.
26	63.	438.	63.	203.	240.	277.	314.	352.	389.	426.	464.	501.	538.	576
27	63.	426.	65.	161.	202.	243.	284.	325.	366.	407.	448.	489.	530.	571.
28	62.	416.	65.	165.	203.	242.	281.	320.	358.	397.	436.	474.	513.	552.
29	63.	368.	57.	167.	202.	238.	273.	308.	343.	379.	414.	449.	485.	520.
30	63.	302.	42.	160.	193.	227.	261.	295.	328.	362.	395.	429.	463.	497.
31	63.	369.	74.	139.	170.	201.	232.	264.	295.	326.	357.	388.	419.	450.
32	63.	307.	59.	148.	175.	202.	230.	257.	284.	311.	339.	366.	393.	421.
33	63.	313.	69.	153.	176.	199.	222.	245.	268.	292.	315.	338.	361.	384.
34	63.	231.	48.	97.	125.	153.	181.	209.	237.	265.	293.	321.	349.	377.
35	63.	212.	48.	97.	121.	145.	169.	193.	217.	241.	265.	289.	313.	338.
36	62.	178.	43.	81.	104.	127.	149.	172.	195.	217.	240.	263.	285.	308.
37	63.	178.	49.	77.	98.	119.	139.	160.	181.	202.	223.	243.	264.	285.
38	62.	125.	28.	63.	85.	108.	130.	153.	176.	198.	221.	243.	266.	288.
39	63.	150.	46.	60.	80.	99.	119.	138.	158.	177.	197.	217.	236.	256.
40	63.	137.	45.	67.	82.	98.	114.	130.	145.	161.	177.	193.	208.	224.
41	63.	129.	36.	65.	83.	100.	118.	136.	153.	171.	189.	206.	224.	242.
42	63.	147.	50.	66.	82.	98.	115.	131.	147.	163.	179.	195.	211.	227.
43	63.	122.	34.	67.	83.	100.	116.	132.	148.	165.	181.	197.	213.	230.
44	63.	135.	40.	65.	83.	100.	117.	135.	152.	169.	186.	204.	221.	238.
45	66.	149.	45.	69.	87.	105.	122.	140.	158.	175.	193.	211.	228.	246.
46	70.	166.	49.	67.	87.	107.	127.	147.	167.	187.	207.	227.	247.	267.
47	70.	180.	44.	93.	113.	133.	153.	173.	192.	212.	232.	252.	272.	291.
48	67.	194.	44.	106.	126.	146.	166.	185.	205.	225.	245.	265.	285.	305.
49	57.	206.	43.	110.	132.	155.	177.	199.	221.	244.	266.	288.	310.	332.
50	55.	200.	38.	100.	126.	152.	178.	204.	230.	256.	282.	308.	334.	360.
51	56.	273.	50.	132.	160.	189.	217.	245.	273.	301.	329.	357.	385.	413.
52	56.	231.	33.	143.	169.	196.	222.	249.	275.	302.	328.	355.	381.	408.
53	10.	306.	54.	108.	144.	181.	217.	253.	290.	326.	363.	399.	435.	472.

RADIATION STATION LEMONT, ILL. (NO. X5023) JANUARY 1957-JUNE 1966
SUNSHINE STATION CHICAGO, ILL. (NO. 14819)

WEEK NO.	95 PERCENT CONFIDENCE LIMITS FOR PREDICTED MEAN RADIATION FOR EACH 10 PERCENT INCREMENT OF MEASURED SUNSHINE										
	0	10	20	30	40	50	60	70	80	90	100
1	131.	131.	130.	130.	130.	130.	130.	130.	131.	132.	132.
2	156.	155.	154.	154.	154.	154.	154.	155.	156.	157.	158.
3	176.	175.	174.	174.	173.	173.	173.	173.	174.	175.	175.
4	176.	175.	174.	173.	173.	173.	173.	173.	173.	174.	175.
5	143.	142.	141.	141.	140.	140.	140.	141.	141.	142.	142.
6	123.	123.	122.	122.	121.	121.	122.	122.	122.	123.	123.
7	226.	225.	223.	222.	222.	221.	221.	221.	222.	222.	223.
8	138.	137.	136.	136.	136.	136.	136.	136.	136.	137.	137.
9	166.	165.	164.	164.	163.	163.	163.	163.	163.	164.	164.
10	166.	165.	163.	162.	161.	161.	161.	161.	161.	161.	162.
11	141.	140.	139.	139.	138.	138.	137.	137.	138.	138.	139.
12	162.	161.	160.	159.	159.	158.	158.	158.	158.	159.	159.
13	193.	190.	188.	186.	185.	184.	183.	183.	183.	184.	185.
14	185.	183.	181.	180.	179.	178.	178.	178.	178.	179.	180.
15	138.	137.	136.	135.	135.	135.	134.	134.	135.	135.	136.
16	148.	146.	145.	143.	142.	142.	141.	141.	141.	142.	142.
17	161.	159.	157.	156.	154.	153.	153.	153.	153.	153.	154.
18	243.	241.	239.	237.	236.	235.	235.	235.	235.	236.	237.
19	190.	188.	187.	185.	184.	183.	183.	183.	183.	183.	184.
20	220.	218.	216.	214.	213.	212.	212.	212	212.	213.	214.
21	214.	211.	209.	207.	205.	204.	203.	203.	203.	204.	205.
22	157.	156.	154.	153.	153.	152.	152.	152.	152.	153.	154.
23	151.	149.	148.	148.	147.	146.	146.	146.	147.	147.	148.
24	162.	161.	159.	158.	157.	157.	156.	156.	157.	157.	158.
25	116.	115.	114.	113.	112.	111.	111.	111.	111.	112.	112.
26	152.	150.	149.	148.	147.	147.	146.	146.	147.	147.	148.
27	114.	113.	112.	111.	111.	110.	110.	110.	110.	111.	111.
28	116.	115.	114.	114.	113.	113.	112.	112.	113.	113.	114.
29	122.	121.	120.	120.	120.	119.	119.	120.	120.	120.	121.
30	141.	141.	140.	140.	140.	140.	140.	140.	141.	142.	143.
31	118.	117.	115.	114.	114.	113.	113.	113.	113.	113.	113.
32	162.	161.	160.	160.	159.	159.	159.	159.	160.	160.	161.
33	88.	87.	87.	86.	86.	86.	86.	85.	86.	86.	86.
34	118.	118.	117.	117.	117.	117.	117.	117.	118.	118.	119.
35	80.	80.	80.	80.	79.	79.	79.	80.	80.	80.	80.
36	80.	80.	80.	80.	80.	80.	80.	80.	80.	81.	81.
37	76.	76.	75.	75.	75.	75.	75.	75.	76.	76.	76.
38	68.	68.	68.	68.	68.	68.	68.	69.	69.	70.	70.
39	67.	67.	67.	67.	67.	67.	67.	67.	67.	67.	68.
40	65.	64.	64.	64.	64.	64.	64.	64.	64.	65.	65.
41	75.	75.	74.	74.	74.	74.	75.	75.	75.	76.	76.
42	77.	76.	76.	76.	76.	76.	76.	76.	76.	76.	77.
43	75.	75.	75.	75.	75.	75.	75.	75.	75.	76.	76.
44	66.	66.	66.	65.	65.	65.	66.	66.	66.	66.	67.
45	68.	68.	68.	67.	67.	67.	67.	68.	68.	68.	68.
46	73.	72.	72.	72.	72.	72.	72.	72.	72.	72.	73.
47	100.	99.	99.	99.	99.	99.	99.	99.	99.	100.	100.
48	83.	82.	82.	82.	82.	82.	82.	82.	82.	83.	83.
49	92.	92.	91.	91.	91.	91.	91.	91.	92.	92.	93.
50	99.	99.	99.	98.	98.	98.	99.	99.	99.	100.	100.
51	94.	94.	93.	93.	93.	93.	93.	93.	93.	94.	94.
52	130.	129.	129.	129.	129.	129.	129.	130.	130.	131.	132.
53	159.	156.	153.	150.	149.	148.	148.	149.	150.	153.	156.

RADIATION STATION LEMONT, ILL, (NO. X5023) JANUARY 1957-JUNE 1966
SUNSHINE STATION CHICAGO, ILL. (NO. 14819)

CHANCE OF HAVING AT LEAST THE INDICATED PERCENT POSSIBLE SUNSHINE****AND AT LEAST INDICATED TOTALS OF LANGLEYS

WEEK	10	20	30	40	50	60	70	80	90	(PERCENT ACCUMULATED)XXXX	100	200	300	400	500	600	700	800(LANGLEYS)
1	68.	64.	58.	49.	47.	41.	34.	27.	19.		85.	61.	44.	24.	2.	0	0	0
2	68.	62.	59.	46.	40.	35.	24.	16.	6.		87.	70.	51.	21.	2.	0	0	0
3	81.	76.	70.	65.	57.	46.	40.	32.	22.		89.	73.	60.	40.	14.	0	0	0
4	79.	75.	68.	67.	59.	54.	44.	32.	19.		89.	78.	57.	46.	14.	0	0	0
5	85.	78.	72.	64.	57.	49.	39.	28.	15.		93.	78.	63.	42.	19.	3.	0	0
6	84.	73.	63.	57.	47.	41.	33.	23.	19.		93.	76.	56.	41.	27.	11.	0	0
7	90.	86.	76.	71.	63.	59.	50.	36.	17.		96.	83.	76.	59.	39.	14.	1.	0
8	86.	80.	73.	63.	56.	49.	39.	33.	26.		93.	83.	70.	57.	39.	21.	0	0
9	82.	79.	73.	66.	64.	58.	54.	40.	33.		96.	78.	69.	57.	51.	36.	3.	0
10	92.	90.	84.	75.	71.	67.	57.	49.	33.		97.	90.	83.	60.	54.	32.	10.	0
11	87.	85.	81.	76.	71.	65.	55.	45.	37.		94.	84.	77.	66.	56.	40.	8.	0
12	84.	81.	78.	76.	71.	62.	54.	48.	33.		100.	87.	76.	68.	60.	40.	11.	0
13	95.	94.	90.	84.	79.	71.	56.	44.	27.		98.	97.	90.	79.	67.	41.	13.	0
14	96.	94.	85.	81.	74.	63.	57.	38.	28.		100.	100.	93.	84.	68.	41.	18.	3.
15	93.	87.	86.	76.	64.	56.	43.	34.	21.		99.	93.	87.	77.	59.	39.	14.	0
16	97.	96.	89.	86.	80.	71.	59.	49.	29.		100.	96.	91.	87.	70.	53.	24.	0
17	99.	96.	94.	89.	79.	70.	57.	50.	36.		100.	100.	97.	93.	73.	56.	26.	1.
18	94.	91.	83.	74.	67.	64.	56.	47.	30.		98.	97.	92.	79.	68.	48.	18.	3.
19	95.	92.	84.	79.	75.	68.	59.	52.	40.		98.	94.	86.	76.	67.	54.	21.	0
20	95.	92.	87.	79.	67.	62.	48.	35.	21.		98.	92.	87.	75.	63.	41.	6.	0
21	97.	95.	90.	83.	76.	65.	60.	51.	32.		100.	97.	89.	83.	73.	51.	8.	2.
22	90.	84.	83.	79.	68.	65.	51.	37.	21.		100.	92.	83.	73.	60.	29.	2.	0
23	94.	83.	81.	75.	70.	56.	51.	35.	30.		100.	95.	89.	79.	60.	30.	2.	0
24	98.	95.	87.	74.	63.	53.	48.	35.	27.		100.	97.	90.	74.	58.	15.	0	0
25	97.	94.	87.	83.	75.	65.	51.	41.	25.		100.	98.	87.	73.	46.	8.	0	0
26	94.	89.	84.	79.	70.	62.	52.	41.	22.		97.	92.	86.	75.	37.	3.	0	0
27	89.	87.	84.	78.	71.	67.	56.	33.	25.		100.	89.	83.	67.	40.	0	0	0
28	90.	89.	82.	79.	74.	65.	53.	40.	24.		97.	90.	81.	66.	27.	0	0	0
29	83.	79.	68.	65.	59.	56.	48.	35.	27.		94.	86.	75.	48.	22.	0	0	0
30	75.	60.	52.	51.	40.	35.	30.	24.	11.		87.	76.	56.	29.	5.	0	0	0
31	94.	94.	86.	83.	78.	75.	68.	57.	44.		98.	90.	76.	57.	0	0	0	0
32	86.	78.	73.	60.	59.	52.	46.	44.	37.		92.	78.	62.	25.	0	0	0	0
33	89.	81.	79.	76.	75.	71.	65.	59.	44.		94.	89.	67.	14.	0	0	0	0
34	73.	70.	60.	59.	54.	41.	37.	29.	21.		79.	67.	38.	0	0	0	0	0
35	65.	63.	56.	51.	49.	46.	38.	35.	27.		81.	54.	33.	0	0	0	0	0
36	65.	56.	53.	47.	39.	34.	34.	31.	26.		69.	48.	15.	0	0	0	0	0
37	67.	67.	65.	56.	51.	46.	38.	33.	27.		76.	49.	2.	0	0	0	0	0
38	52.	45.	39.	32.	27.	19.	16.	10.	6.		55.	21.	0	0	0	0	0	0
39	71.	65.	54.	52.	48.	44.	38.	30.	16.		70.	40.	0	0	0	0	0	0
40	65.	62.	59.	51.	46.	40.	37.	24.	17.		67.	22.	0	0	0	0	0	0
41	59.	49.	41.	40.	38.	30.	27.	21.	17.		63.	19.	0	0	0	0	0	0
42	68.	62.	62.	59.	56.	48.	38.	35.	29.		70.	33.	0	0	0	0	0	0
43	52.	44.	40.	35.	33.	30.	27.	24.	16.		56.	24.	0	0	0	0	0	0
44	68.	54.	51.	46.	41.	35.	32.	22.	14.		70.	24.	0	0	0	0	0	0
45	71.	64.	59.	50.	42.	38.	32.	30.	26.		71.	38.	0	0	0	0	0	0
46	73.	69.	63.	60.	54.	47.	37.	24.	23.		77.	46.	3.	0	0	0	0	0
47	60.	57.	56.	49.	46.	44.	43.	30.	17.		70.	57.	4.	0	0	0	0	0
48	76.	69.	60.	48.	42.	36.	30.	24.	21.		81.	54.	12.	0	0	0	0	0
49	70.	61.	53.	49.	46.	39.	33.	26.	16.		82.	49.	18.	0	0	0	0	0
50	55.	53.	49.	44.	42.	33.	27.	25.	20.		73.	51.	24.	0	0	0	0	0
51	73.	64.	61.	59.	55.	46.	41.	32.	27.		93.	66.	50.	16.	0	0	0	0
52	57.	52.	41.	34.	30.	27.	23.	18.	16.		82.	63.	30.	11.	0	0	0	0
53	70.	60.	60.	60.	60.	60.	50.	50.	30.		80.	70.	60.	50.	0	0	0	0

RADIATION STATION LEMONT, ILL. (NO. X5023) JANUARY 1957-JUNE 1966
SUNSHINE STATION CHICAGO, ILL. (NO. 14819)